2022－10－11

第３章　これならできるコマのシミュレーション＜エクセル VBA＞

東亜天文学会　江頭　務

はじめに

シミュレーションというと難解なプログラム言語を駆使するようなイメージがありますが、エクセル VBA を使えば誰でも比較的容易にプログラムを作成することができます。
ここでは、エクセル VBA によるコマのシミュレーションプログラムを紹介いたします。
本稿のプログラムをコピーしてエクセル VBA の標準モジュールに張り付けるだけで、誰でも簡単にコマのシミュレーションを行うことができます。
VBA の基本的な事項は入門書や Web 等を参照願います。
本稿は『理科教材で学ぶコマの運動』の一部で、説明で引用する考え方や諸式はすべて下記のサイトにまとめて掲載しています。

コマのワンダーランド　理科教材で学ぶコマの運動＜歳差・章動・極運動＞
https://yamauo1945.sakura.ne.jp/astronomy.html

１ シミュレーションの流れのポイント

1　一般のエクセル画面（ファイル拡張子.xlsx）を立ち上げ
2　入力画面の作成
3　開発タブの表示
4　Visual Basic をクリック
5　プロジェクトの下に標準モジュールを挿入
6　VBA の貼り付け
7　マクロ実行ボタンの追加（繰り返しシミュレーションを実行するのに便利）
8　シミュレーションの実行（グラフの作成方法は一般のエクセルと同様）
9　名前をつけてマクロ有効ブック（ファイル拡張子. xlsm ）にて保存（拡張子.xlsx を必ず変更すること。

＜基本方針＞
VBA の利点は繰り返し計算にあるといっても良いだろう。
プログラムのデバッグを容易にするには、できるだけ VBA にしかできないことに限定してプログラムを簡素化するのが賢明である。
そのため、単純計算はできるだけ通常のエクセル画面で行うこととした。

2　Excel VBA 入力画面

図 3.1 にエクセル入力画面を示す。

用語や記号の詳細については、１章「単振動近似」と２章「章動一般」を参照願いたい。

	A	B	C	D	E	F	G	H	I
1	計算条件			初期条件			計算結果グラフ		
2	積分区間_init	0		t0=	0	s	時間 t	秒 s	J列
3	積分区間_end	5		x0(θ_1)=	1.0472	rad	変位θ	rad表示	K列
4	刻み幅h	0.001		v0(θdot)=	0		速度θdot	rad表示	L列
5	出力幅h2	10		p0=	0		加速度θ2dot	rad表示	M列
6	計算回数	5000		u0=	0		変位Φ	rad表示	N列
7	コマの基本条件						速度Φdot	rad表示	O列
8	安定度　S	6					変位φ	度表示	P列
9	mgr(g=9.8)	2.94					変位θ	度表示	Q列
10	I	0.135					速度θdot	度表示	R列
11	I_3	0.09					速度φdot	回転/s	S列
12	L_3	1.543					コマ速度	回転/s	T列
13	$\omega_3=L_3/I_3$ w3	17.144					コマ回転	回転数	U列
14							加速度φ2dot	rad表示	V列
15	コマの運動						コマ角加速度	rad表示	W列
16	傾斜角θrad	1.047198	60	度	z係数比率		正距方位距離	度表示	X列
17	付加的角運動量La	-0.233595	-0.15139	z係数	-1.1	Rz			
18	mgzr	-1.780758	-0.6057	gz(単位g)					
19							koma4		
20	S=6 1g で定常歳差運動を行う時の z_c係数を求める								
21	計算1	計算2	z_{c1}				計算カウンター	5000	
22	0.75	0.816497	0.137628						
23									
24	正距方位図	極運動半径		章動幅中央値					
25		rad	度	rad	度				
26	基準点θ座標rad	-0.173	-9.916	1.220	69.916				
27									
28	歳差とコマの回転速度　回転数/s								
29		歳差	コマ						
30	最大値	-0.000145	2.912147						
31	最小値	-0.367189	2.728603						
32	平均値	-0.173537	2.808894	回転数/時間					

図 3.1　エクセル入力画面

注

1 入力画面の数字が記載されたセルの位置は、VBA とリンクしているので位置を変えないこと。

2 出力は J～W 列の 1 行目から B6 の回数分まで行われる。

3 グラフは J～W 列のアルファベットの部分をクリックすれば範囲指定ができる。

＜図 3.1 エクセル入力画面の解説＞

[A1：B6 ブロック］数値積分の計算条件

積分区間は B2 の 0 秒から B3 の 5 秒の間である。B3 の 5 秒は変更することができる。
B 4 は数値積分の刻み幅で 1/1000 秒である。これも特に細かく計算したいところでは刻み幅を短くすることができる。
B5 の出力幅は、グラフのプロット数に相当するもので計算回数 5000 回の 1/10 とした。
B6 の計算回数は積分区間 B3 を刻み幅 B4 で除したものである。

[A7：B13 ブロック］コマの基本条件

B8～B13 はコマの基本条件を示すもので、検討の対象となるコマの状態を下記に示す。

表 3.1 ＜検討対象となるコマの状態＞

	Lz:鉛直軸の角運動量 第 3 軸:コマの慣性主軸 コマの軸と同じ 第 1 軸、第 2 軸： 第 3 軸に直交する 慣性主軸 L_3：コマの軸の角運動量 ω_3：第 3 軸の角速度	$\dot{\phi}$：歳差運動の角速度 θ ：コマの軸の傾斜角 g ：重力加速度 m ：コマの質量 G: コマの重心 r ：位置ベクトル O: コマの支点（原点）

地球上で安定度 $S=L_3^2/mgrI=6$（安定度 S については１章「単振動近似」参照）のコマが軸の傾斜角 $\theta_1=60°$ で定常歳差運動している状態

質量 m＝1 kg　重力加速度　$g=9.8m/s^2$　r=0.3m

コマの回転軸に直交する軸周りの慣性モーメント $I=0.135kgm^2$

回転軸周りの慣性モーメント $I_3=0.09kgm^2$　　第 3 軸の角運動量 $L_3=1.543kgm^2/s$

B9 は mgr=1×9.8×0.3=2.94、B13 は第 3 軸の角速度 $\omega_3= L_3/I_3=1.543/0.09=17.144$、
これらは定数である。

[D1：E6 ブロック］コマの初期条件

それぞれ時間 t=0 における値で、t0 時間 0s、x0(θ_1)コマの傾斜角 θ rad である。

v0(θ dot)は、コマの傾斜角の角速度$\dot{\theta}$で初期値は 0rad/s である。

これは、コマの傾斜角 θ の初期値が章動の波高値になることによる。

また p0 は歳差運動 ϕ の初期位置で 0rad、u0 はコマの回転角φの初期位置で 0rad に設定した。

[G1：I16 ブロック］計算結果グラフ

エクセルの J 列～W 列のところに B5 の出力幅に対応した計算結果が表示される。

任意の列を組み合わせて、グラフを作成することができる。

時間 t は秒。コマの傾斜角の変位 θ 、角速度$\dot{\theta}$、角加速度$\ddot{\theta}$、歳差の変位 ϕ 、角速度$\dot{\phi}$は rad で表示。次の ϕ 、 θ 、$\dot{\theta}$、$\dot{\phi}$は度で表示。

続いて$\dot{\phi}$とコマの速度$\dot{\varphi}$は秒あたりの回転数で表示した。

次のコマ回転とあるのは B3 の計算時間に対応したコマの総回転数φである。

次の加速度 ϕ 2dot とあるのは歳差の角加速度$\ddot{\phi}$、次のコマ角加速度とあるのはコマの角加速度$\ddot{\varphi}$で、rad で表示してある。

最後の G16 の正距方位距離とあるのは、無重力状態のコマにおいては円運動が起こるため中心点からの角距離が一定であることを数値的に確認するためのものである。

ϕ 、 θ は度で表示した。(詳細第 6 章参照)

[A15：F32 ブロック］コマの運動

C16 ここにコマの傾斜角 θ（度表示）を入力する。B16 はそれを rad に換算したものでこの値が VBA に入力される。

E17 ここに z 係数比率 Rz を入力する。

z 係数比率 Rz は図 2.1 の歳差と章動によって形成されるコマの先端の運動軌跡の条件 (a) $L_z > L_3\cos\theta_1$ (b) $L_z = L_3\cos\theta_1$ (c) $L_z < L_3\cos\theta_1$ をコントロールするもので、

(a) $Rz > 0$ 、(b) $Rz = 0$ 、(c) $Rz < 0$ が対応する。

C17 は Rz を z 係数に換算したもの、B17 は z 係数をさらに付加的角運動量 La に換算したものでこの値が VBA に入力される。

付加的運動量 La は、$\mathrm{L_a = L_3 z = L_3 z_{c1} R_z} \quad \mathrm{L_z = L_3(\cos\theta_1 + z) = L_3\cos\theta_1 + L_a}$ (2.16)式

$z_{c1} = \frac{\sin^2\theta_c}{2\cos\theta_c}\left\{1 - \sqrt{1 - \frac{4\cos\theta_c}{S}}\right\}$ (2.10)式　である。

VBA には La が入力されるので、z 係数の計算を飛ばして B17 に直接入力することも可能である。

C18 は一般化された重力加速度 g_z で 1g(9.8m/s^2)を単位とする。
B18 はこれをもとに $mg_z r$ を計算したものでこの値が VBA に入力される。

C22 は、S=6　1g で定常歳差運動を行う時の z_{c1} 係数である。
A22,B22 は z_{c1} 係数を求めるための計算ステップ 1 と 2 の値である。
この z_{c1} 係数に E17 の z 係数比率 Rz を掛けたものが C17 の z 係数となる。
B17 の付加的角運動量 La は、B12 の L_3 に C17 の z 係数を乗じたものとなる。
これらを式であらわせば、前述の $\mathrm{L_a = L_3 z = L_3 z_{c1} R_z}$　(2.16)式となる。

A24：E26 は、章動の変動幅の中央値を求めるためのものである。
変動幅の中央値($\theta_1 - \theta_2$)/2 は、$\tan^{-1}(La/L_3 \sin\theta_1)$　(6.46)式で求められる。
La が正の時、中央値は θ_1 を基準にして減少方向、負の時は増大方向に作用することに留意すること。これは主として、無重力下のコマの円運動を確認するために用いられる。
最終的に D26 の rad 値が VBA に入力される。(詳細第 6 章参照)

A28～D32 は歳差とコマの回転速度の最大値、最小値、平均値を秒あたりの回転数で表示したものである。
最大値と最小値の検出はエクセルの MAX 関数と MIN 関数を使用した。
エクセル画面の最大行数は（B6 の計算回数/B5 の出力幅）+1 である。
歳差とコマの回転速度の平均値は、計算の完了時点でのコマの回転数を計算時間で除したものである。尚、回転数はコマが右回転する場合は、負の回転として計算される。
コマが左右に等しく回転すれば、回転数は 0 となる

[G19：H24 ブロック]計算実行ボタンと計算カウンター

H21 に計算のステップがリアルタイムで表示される。
その上にあるのは、計算実行ボタン。

3　Excel VBA

下記のプログラムを、コピーして標準モジュールに貼り付ける。

```
Public An, L3, Ir, La, w3, mgzr As Double
Sub koma4()   'T.Egashira  公式   2022-10-11

Dim init, ed, h, h2, x0, v0, p0, u0, kx(4), kv(4), kp(4), ku(4) As Double
Dim i, j As Integer

init = Cells(2, 2): ed = Cells(3, 2): h = Cells(4, 2): h2 = Cells(5, 2)

'定数
An = Cells(16, 2): L3 = Cells(12, 2): Ir = Cells(10, 2): La = Cells(17, 2): w3 = Cells(13, 2): mgzr =
Cells(18, 2)

'初期値
x0 = Cells(3, 5): v0 = Cells(4, 5): x = x0: v = v0: t = init
p0 = Cells(5, 5): p = p0 'φ=0:u0 = Cells(6, 5): u = u0 'コマの回転数の初期値 0
k = F2(t, x0, v0) 'θ2dot の初期値
g = F3(t, x0) 'φdot の初期値
q = F4(t, x0, g) 'コマの角速度の初期値
a = F5(v0, x0) 'φ2dot の初期値
b = F6(g, v0, x0, a) 'コマの角加速度の初期値

Range("J:X").ClearContents 'J 列～X 列の数値を clear する

For i = 0 To ((ed - init) / h)
j = 1 + i / h2
Cells(21, 8) = i
If i Mod h2 = 0 Then
Cells(j, 10) = t: Cells(j, 11) = x: Cells(j, 12) = v
Cells(j, 13) = k: Cells(j, 14) = p: Cells(j, 15) = g: Cells(j, 19) = g / 6.2831853: Cells(j, 20) = q /
6.2831853: Cells(j, 21) = u / 6.2831853: Cells(j, 22) = a: Cells(j, 23) = b

Cells(j, 16) = Application.WorksheetFunction.Degrees(p)
```

```
Cells(j, 17) = Application.WorksheetFunction.Degrees(x)
Cells(j, 18) = Application.WorksheetFunction.Degrees(v)
d1 = Application.WorksheetFunction.Acos(Cos(Cells(j, 11)) * Cos(Cells(26, 4)) + Sin(Cells(j, 11)) *
Sin(Cells(26, 4)) * Cos(Cells(j, 14)))　'正距方位図　距離（rad）
d2 = Application.WorksheetFunction.Degrees(d1) '正距方位図　距離（度）
Cells(j, 24) = d2

End If

kx(1) = h * F1(t, x, v)
kv(1) = h * F2(t, x, v)
kp(1) = h * F3(t, x)
ku(1) = h * F4(t, x, kp(1) / h)

kx(2) = h * F1(t + h / 2, x + kx(1) / 2, v + kv(1) / 2)
kv(2) = h * F2(t + h / 2, x + kx(1) / 2, v + kv(1) / 2)
kp(2) = h * F3(t + h / 2, x + kx(1) / 2)
ku(2) = h * F4(t + h / 2, x + kx(1) / 2, kp(2) / h)

kx(3) = h * F1(t + h / 2, x + kx(2) / 2, v + kv(2) / 2)
kv(3) = h * F2(t + h / 2, x + kx(2) / 2, v + kv(2) / 2)
kp(3) = h * F3(t + h / 2, x + kx(2) / 2)
ku(3) = h * F4(t + h / 2, x + kx(2) / 2, kp(3) / h)

kx(4) = h * F1(t + h, x + kx(3), v + kv(3))
kv(4) = h * F2(t + h, x + kx(3), v + kv(3))
kp(4) = h * F3(t + h, x + kx(3))
ku(4) = h * F4(t + h, x + kx(3), kp(4) / h)

nx = x + (kx(1) + 2 * kx(2) + 2 * kx(3) + kx(4)) / 6
nv = v + (kv(1) + 2 * kv(2) + 2 * kv(3) + kv(4)) / 6
np = p + (kp(1) + 2 * kp(2) + 2 * kp(3) + kp(4)) / 6
nu = u + (ku(1) + 2 * ku(2) + 2 * ku(3) + ku(4)) / 6
nt = t + h

t = nt: x = nx: v = nv:nk = F2(t, nx, nv): ng = F3(t, nx): nq = F4(t, nx, ng):
```

```
k = nk: g = ng: p = np: q = nq: u = nu:
na = F5(nv, nx): nb = F6(ng, nv, nx, na)
a = na: b = nb

Next
End Sub

Function F1(ByVal t As Double, ByVal x As Double, ByVal v As Double) As Double
F1 = v 'θdot
End Function

Function F2(ByVal t As Double, ByVal x As Double, ByVal v As Double) As Double
F2 = (mgzr * Sin(x) - ((La + L3 * Cos(An) - L3 * Cos(x)) / Sin(x) ^ 2) * (L3 - ((La + L3 * Cos(An) -
L3 * Cos(x)) / Sin(x) ^ 2) * Cos(x)) * Sin(x) / Ir) / Ir 'θ2dot
End Function

Function F3(ByVal t As Double, ByVal x As Double) As Double
F3 = (La + L3 * Cos(An) - L3 * Cos(x)) / (Ir * (Sin(x) ^ 2)) 'φdot
End Function

Function F4(ByVal t As Double, ByVal x As Double, ByVal g As Double) As Double
F4 = -g * Cos(x) + w3 'コマの角速度
End Function

Function F5(ByVal v As Double, ByVal x As Double) As Double
F5 = (v / (Ir * Sin(x) ^ 3)) * (L3 * (1 + Cos(x) ^ 2) - 2 * (L3 * Cos(An) + La) * Cos(x)) 'φ2dot
End Function

Function F6(ByVal g As Double, ByVal v As Double, ByVal x As Double, ByVal a As Double) As
Double
F6 = g * v * Sin(x) - a * Cos(x) 'コマの角加速度
End Function
```

4　プログラム解説

プログラムの計算は4次精度ルンゲ・クッタ法にて実行する。
ルンゲクッタ法については下記のサイトがわかりやすい。
「数値計算を使って常微分方程式を解く～ルンゲクッタ法の解説～」
http://shimaphoto03.com/science/rk-method/

F1 = v 'θ dot（コマの軸の傾斜角 θ の角速度$\dot{\theta}$）

F2 = (mgr * Sin(x) - ((La + L3 * Cos(An) - L3 * Cos(x)) / Sin(x) ^ 2) * _
(L3 - ((La + L3 * Cos(An) - L3 * Cos(x)) / Sin(x) ^ 2) * Cos(x)) * _
Sin(x) / Ir) / Ir 'θ 2dot

x= θ An= θ_1（初期条件におけるコマの軸の傾斜角　定数）

角加速度$\ddot{\theta}$を導く(2.2)式をコード化したもの。

$$\ddot{\theta} = -\left(\frac{L_3}{I}\right)^2 \left\{\left(\frac{(cos\theta_1 + z) - cos\theta}{sin^2\theta}\right)\left(1 - \left(\frac{(cos\theta_1 + z) - cos\theta}{sin^2\theta}\right) cos\theta\right) sin\theta\ - \frac{1}{S}\left(\frac{g_z}{g}\right) sin\theta\right\} \quad (2.2)式$$

F3 = (La + L3 * Cos(An) - L3 * Cos(x)) / (Ir * (Sin(x) ^ 2)) 'ϕ dot

歳差運動の角速度$\dot{\phi}$を導く(2.1)式をコード化したもの。

$$\dot{\phi} = \frac{1}{I\sin^2\theta}(L_z - L_3\cos\theta) \quad (2.1)式$$

F4 =w3 -g * Cos(x) 'ω 3dot　w3=ω_3（安定度Sより求まるω_3　定数）　g= $\dot{\phi}$

コマの回転速度$\dot{\varphi}$を求める式$\dot{\varphi}=\omega_3-\dot{\phi}\cos\theta$（1.2)式　をコード化したもの。

F5 = (v / (Ir * Sin(x) ^ 3)) * (L3 * (1 + Cos(x) ^ 2) - 2 * (L3 * Cos(An) + La) * Cos(x)) ' φ2dot 歳差の角加速度$\ddot{\phi}$を求める式

$$\ddot{\phi} = \frac{\dot{\theta}}{I\sin^3\theta}(L_3(1+\cos^2\theta) - 2\,L_z\cos\theta) \quad (2.25)式をコード化したもの。$$

F6 = g * v * Sin(x) - a * Cos(x) ' コマの角加速度を求める式
$\ddot{\varphi} = \dot{\phi}\dot{\theta}sin\theta - \ddot{\phi}\cos\theta$　(2.25)式をコード化したもの。

無重力下のコマの章動の中心点からの角距離を求めるための数式である。
d1 = Application.WorksheetFunction.Acos(Cos(Cells(j, 11)) * Cos(Cells(26, 5)) + Sin(Cells(j, 11)) * Sin(Cells(26, 5)) * Cos(Cells(j, 14)))　'正距方位図　距離（rad)

$cosd = cos\theta_1 cos\theta_2 + sin\theta_1 sin\theta_2 cos(\phi_2 - \phi_1)$　(6.44)式をコード化したもの。

（詳細第6章参考2参照）